
Besvarelsen skal gis som programkode (Visual Basic for Applications). 

1 Salgsoversikt
Et forlag har et datasystem som gir oversikt over alle salg. En tabell Salg inneholder feltene:

· Dato 
· Tittel
· Antall bøker solgt
1.1 Antall bøker solgt
Vi har et skjema der det er en komboboks der en velger bok. Beskriv SQL setningen for å velge ut titler fra tabellen Salg. 
SELECT DISTINCT Salg.tittel FROM Salg ORDER BY Salg.[tittel];

Skjemaet har en knapp med tekst Antall, som teller opp det totale antall bøker solgt. Vis koden som hører til ”on-click eventet” for denne knappen. Svaret, antall bøker solgt, kan gis i en melding.

Private Sub comAntall_Click()

    Dim stdset As DAO.Recordset

    Dim strsql As String

    Dim sum As Long

    strsql = "SELECT * FROM Salg WHERE tittel='" & cmbTittel & "'"

    Set stdset = CurrentDb.OpenRecordset(strsql)

    With stdset

        sum = 0

        Do While Not .EOF

’Alternativt sum=sum + 1 om en vil telle antall salg og ’ikke antall eksemplarer

            sum = sum + !antall 

            .MoveNext

        Loop

        .Close

    End With

    Call MsgBox("antall salg: " & sum)

End Sub
1.2 Antall større salg

Skjemaet har en knapp Større salg. Her skal vi kun telle opp salg der antallet er 2 eller høyere. Programmer koden for dette.
strsql = "SELECT * FROM Salg WHERE tittel='" & cmbTittel & "' AND _ antall > 1"

For øvrig som over.
1.3 Inntekter

I systemet ligger en tabell Prisliste med følgende felt:
· Tittel

· Pris
I grensesnittet er det en knapp for Finn sum. Når denne trykkes skal systemet regne ut summen i kroner av alle salg for den oppgitte boken. Programmer dette. Svaret (summen) gis som en melding.

Private Sub comInntekter_Click()

    Dim stdset As DAO.Recordset, prisSet As DAO.Recordset

    Dim strsql As String

    Dim sum As Long

    'dette kan løses ved å lage en query der vi "joiner" salg og 

    'SELECT Prisliste.tittel, Prisliste.pris, Salg.antall FROM

    'Prisliste INNER JOIN Salg ON Prisliste.tittel = Salg.tittel;

    'Med denne kan vi hente frem alt vi trenger fra queryen, men vi
    'har her valgt å programmere løsningen

    strsql = "SELECT * FROM Salg WHERE tittel='" & cmbTittel & "' AND antall > 1"

    Set stdset = CurrentDb.OpenRecordset(strsql)

    With stdset

        sum = 0

        Do While Not .EOF

            strsql = "SELECT pris FROM Prisliste WHERE tittel ='" & !tittel & "'"

            Set prisSet = CurrentDb.OpenRecordset(strsql)

            sum = sum + !antall * prisSet!pris

            prisSet.Close

            .MoveNext

        Loop

        .Close

    End With

    Call MsgBox("totale inntekter: " & sum)

End Sub
1.4 Velg ut år

Beskriv en komboboks der bruker kan velge et årstall (vis SQL setningen som henter årstall fra salg). 
SELECT DISTINCT Year([dato]) AS år FROM Salg ORDER BY Year([dato]);

Løs oppgave 1.1 på nytt, men ta nå kun med bøker solgt i det oppgitte året.

strsql = "SELECT * FROM Salg WHERE tittel='" & cmbTittel & "' AND year(dato) =" & cmbÅr
Ellers som over.

Korrektur

Under innskriving av tekst er det lett å legge inn for mange blanke. Programmer knappen Fjern unødvendige blanke som fjerner alle forekomster av to eller flere blanke etter hverandre.

Private Sub comFjern_Click()

    Dim s As String

    Dim forrige As String

    Dim p As Integer

    'vi går gjennom teksten og legger denne inn i stringen s bokstav for bokstav, så
    'sant forrige bokstav ikke var en blank. Til slutt legger vi s inn i txtTekst

    forrige = Mid(txtTekst, 1, 1)

    s = forrige

    For p = 2 To Len(txtTekst)

        If forrige = " " And Mid(txtTekst, p, 1) = " " Then

            'skip blank

        Else

            s = s & Mid(txtTekst, p, 1)

            forrige = Mid(txtTekst, p, 1)

        End If

    Next p

    txtTekst = s

End Sub
3 Innlesing fra tekstfil

Forlaget i oppgave 1 har en egen avdeling for fagbøker. Disse har sitt eget datasystem med tabellen Salg. Her legger de inn salg i hver måned. Etter at måneden er over sender de innholdet av Salg til hovedkontoret som en tekstfil med navn salgsdata.txt i følgende format:

Altså dato, tittel og antall solgt med en blank mellom hvert datafelt. Programmer en knapp Importer som leser data i tekstfilen (salgsdata.txt) og legger dette inn i tabellen Salg i hovedkontorets system. Du kan gå ut fra at tekstfilen ligger i samme mappe som programmet.
Vil løsningen din fungere om det er blanke tegn i boktittelen, for eksempel ”Programmering VB” og ”Betonglære 1”? Beskriv hvordan du vil løse dette.

Private Sub comImporter_Click()

    Dim fnr As Integer, k As Integer

    Dim linje As String

    Dim dato As String

    Dim tittel As String

    Dim antall As Long

    Dim strsql As String

    fnr = FreeFile

    Open CurrentProject.Path & "\" & "salgsdata.txt" For Input As fnr

    Do While Not EOF(fnr)

        Input #fnr, linje

        linje = Trim(linje) ' fjerner eventuelle blanke foran og bak

        dato = Mid(linje, 1, 8)

        'går baklengs for å finne antallet, da kan vi akseptere blanke i tittel

        For k = Len(linje) To 1 Step -1

            If Mid(linje, k, 1) = " " Then

                Exit For

            End If

        Next k

        antall = Mid(linje, k + 1, Len(linje) - k)

        tittel = Mid(linje, 10, k - 10)

        strsql = "INSERT INTO Salg (dato, tittel, antall) VALUES ('" & dato & "','" & _ 

           tittel & "'," & antall & ")"

        CurrentDb.Execute (strsql)

    Loop

    Close (fnr)

    Call MsgBox("Filen er importert og lagret i Salg")
End Sub

4 Beregning av median
Medianen er det tallet i en sortert sekvens som har like mange (større) tall over seg som (mindre tall) under seg. I sekvensen 2, 5, 9, 11, 14 er medianen 9. Dersom det er et like antall tall i sekvensen vil medianen ligge mellom to tall. Medianen kan da settes som gjennomsnittet av tallet over og under, eller – som vi skal gjøre her – å velge ut et av de midterste tallene som median (det kan være å foretrekke at medianen er et tall som finnes i datasettet). I sekvensen 2, 5, 9, 11, 14, 15 kan vi altså la medianen være 9 eller 11.

For å finne medianen kan vi sortere og så plukke ut det/de midterste tallene. Det skal vi ikke gjøre her. Isteden skal vi starte med et tall i (den usorterte) sekvensen og telle opp antall over og under. Deretter skal vi plukke det neste tallet. Dette skal vi gjøre inntil vi har funnet medianen. 
Vårt forlag (oppgave 1) ønsker å finne medianen av antall bøker solgt i hvert salg. Vi er altså interessert i å finne medianen av feltet Antall i tabellen Salg. 
4.1 Innlesing til array

Programmer en knapp LesInn som leser inn antallet fra tabellen Salg til et array.
Private antall() As Long

Private antallPoster As Long

Private Sub lesInn()

    Dim strsql As String

    Dim stdset As DAO.Recordset

    strsql = "SELECT antall FROM Salg"

    Set stdset = CurrentDb.OpenRecordset(strsql)

    With stdset

        antallPoster = 0

        Do While Not .EOF

            antallPoster = antallPoster + 1

            ReDim Preserve antall(1 To antallPoster)

            antall(antallPoster) = !antall

            .MoveNext

        Loop

        .Close

    End With

End Sub
4.2 Median, odde antall poster, - alle tall ulike

Programmer algoritmen som finner medianen. Her kan du gå ut fra at det er et odde antall poster (”records”) i arrayet og (ikke helt realistisk) at det ikke er like tall i materialet.
Private Sub comMedian_Click()

    Dim p As Long, i As Long

    Dim kandidat As Integer, mindre As Integer, større As Integer, antLike As Integer

    Call lesInn

    For p = 1 To antallPoster

        kandidat = antall(p)

        'test kandidaten

        mindre = 0

        større = 0

        For i = 1 To antallPoster

            If i <> p Then

                If antall(i) < kandidat Then

                    mindre = mindre + 1

                ElseIf antall(i) > kandidat Then

                    større = større + 1

                End If

            End If

        Next i

        If mindre = større Then

            Call MsgBox("Medianen er " & kandidat)

            Exit Sub

        End If

    Next p

    Call MsgBox("Ingen kandidat funnet pga. like tall eller partall antall poster")

End Sub
4.3 Tilpassning til det generelle tilfellet

Se om du kan tilpasse løsningen slik at den også fungerer om det er et partall antall poster og like tall. Kan du programmere dette er det bra, men har du liten tid kan du si litt om hvordan du vil gjøre dette.
Private Sub comGenerell_Click()

    Dim p As Long, i As Long

    Dim kandidat As Integer, mindre As Integer, større As Integer, antLike As Integer

    Call lesInn

    For p = 1 To antallPoster

        kandidat = antall(p)

        'test kandidaten - ta hensyn til at vi kan ha et likt antall poster

        mindre = 0

        større = 0

        antLike = 1  'kandidaten er lik seg selv, starter da med 1

        For i = 1 To antallPoster

            If i <> p Then

                If antall(i) < kandidat Then

                    mindre = mindre + 1

                ElseIf antall(i) > kandidat Then

                    større = større + 1

                Else

                    antLike = antLike + 1

                End If

            End If

        Next i

        If (mindre = større) Or (mindre + antLike) = større Or mindre = (større + _


 antLike) Then

            Call MsgBox("Medianen er " & kandidat)

            Exit Sub

        End If

    Next p

    Call MsgBox("fant ingen median")

End Sub

